

rocregplot — Plot marginal and covariate-specific ROC curves after rocreg
[Description](#)[probit_options](#)[Methods and formulas](#)[Quick start](#)[common_options](#)[References](#)[Menu](#)[boot_options](#)[Also see](#)[Syntax](#)[Remarks and examples](#)

Description

Under parametric estimation, `rocregplot` plots the fitted ROC curves for specified covariate values and classifiers. If `rocreg`, `probit` or `rocreg, probit ml` were previously used, the false-positive rates (for specified ROC values) and ROC values (for specified false-positive rates) for each curve may also be plotted, along with confidence intervals.

Under nonparametric estimation, `rocregplot` will plot the fitted ROC curves using the `_fpr_*` and `_roc_*` variables produced by `rocreg`. Point estimates and confidence intervals for false-positive rates and ROC values that were computed in `rocreg` may be plotted as well.

Quick start

Plot ROC curve after any `rocreg` command

```
rocregplot
```

Name graph `mygraph` and save as `myfile.gph`

```
rocregplot, name(mygraph) saving(myfile)
```

Plot ROC curve only for `v2` after fitting a model with `v1` and `v2`

```
rocregplot, classvars(v2)
```

Add bias-corrected CI for the ROC value at a false-positive rate of 0.7 from estimation with bootstrap resampling that specified `roc(.7)`

```
rocregplot, btype(bc)
```

Plots following parametric estimation only

Plot curve evaluated at `x = 30` and `x = 50`

```
rocregplot, at1(x=30) at2(x=50)
```

After bootstrap resampling with resampled observations saved in `myfile.dta`

```
rocregplot, at1(x=30) at2(x=50) bfile(myfile)
```

And draw CI for ROC values at false-positive rates 0.1, 0.2, and 0.3

```
rocregplot, at1(x=30) at2(x=50) roc(.1 .2 .3) bfile(myfile)
```

Menu

Statistics > Epidemiology and related > ROC analysis > ROC curves after rocreg

Syntax

Plot ROC curve after nonparametric analysis

```
rocregplot [ , common_options boot_options ]
```

Plot ROC curve after parametric analysis using bootstrap

```
rocregplot [ , probit_options common_options boot_options ]
```

Plot ROC curve after parametric analysis using maximum likelihood

```
rocregplot [ , probit_options common_options ]
```

| <i>probit_options</i> | Description |
|---|--|
| Main | |
| <code>at(<i>varname</i>=# [<i>varname</i>=#...])</code> | value of specified covariates and mean of unspecified covariates |
| <code>[at1(<i>varname</i>=# [<i>varname</i>=#...])</code> | |
| <code>[at2(<i>varname</i>=# [<i>varname</i>=#...])</code> | |
| <code>[...]]]</code> | |
| * <code>roc(<i>numlist</i>)</code> | show estimated ROC values for given false-positive rates |
| * <code>invroc(<i>numlist</i>)</code> | show estimated false-positive rates for given ROC values |
| <code>level(#)</code> | set confidence level; default is level(95) |
| Curve | |
| <code>line#opts(<i>cline_options</i>)</code> | affect rendition of ROC curve # |

*Only one of `roc()` or `invroc()` may be specified.

| <i>common_options</i> | Description |
|--|--|
| Main | |
| <code>classvars(<i>varlist</i>)</code> | restrict plotting of ROC curves to specified classifiers |
| <code>norefline</code> | suppress plotting the reference line |
| Scatter | |
| <code>plot#opts(<i>scatter_options</i>)</code> | affect rendition of the classifier #s false-positive rate and ROC scatter points; not allowed with <code>at()</code> |
| Reference line | |
| <code>rlopts(<i>cline_options</i>)</code> | affect rendition of the reference line |
| Y axis, X axis, Titles, Legend, Overall | |
| <code>twoway_options</code> | any options other than <code>by()</code> documented in [G-3] <i>twoway_options</i> |

| <i>boot_options</i> | Description |
|---|--|
| Bootstrap | |
| * <code>bfile(<i>filename</i>)</code> | load dataset containing bootstrap replicates from <code>rocreg</code> |
| <code>btype(n p bc)</code> | plot normal-based (n), percentile (p), or bias-corrected (bc) confidence intervals; default is <code>btype(n)</code> |
| * <code>bfile()</code> is only allowed with parametric analysis using bootstrap inference; in which case this option is required with <code>roc()</code> or <code>invroc()</code> . | |

probit_options

Main

`at(varname=# ...)` requests that the covariates specified by *varname* be set to #. By default, `rocregplot` evaluates the function by setting each covariate to its mean value. This option causes the ROC curve to be evaluated at the value of the covariates listed in `at()` and at the mean of all unlisted covariates.

`at1(varname=# ...)`, `at2(varname=# ...)`, ..., `at10(varname=# ...)` specify that ROC curves (up to 10) be plotted on the same graph. `at1()`, `at2()`, ..., `at10()` work like the `at()` option. They request that the function be evaluated at the value of the covariates specified and at the mean of all unlisted covariates. `at1()` specifies the values of the covariates for the first curve, `at2()` specifies the values of the covariates for the second curve, and so on.

`roc(numlist)` specifies that estimated ROC values for given false-positive rates be graphed.

`invroc(numlist)` specifies that estimated false-positive rates for given ROC values be graphed.

`level(#)` specifies the confidence level, as a percentage, for confidence intervals. The default is `level(95)` or as set by `set level`; see [U] 20.8 Specifying the width of confidence intervals. `level()` may be specified with either `roc()` or `invroc()`.

Curve

`line#opts(cline_options)` affects the rendition of ROC curve #. See [G-3] *cline_options*.

common_options

Main

`classvars`(*varlist*) restricts plotting ROC curves to specified classification variables.
`norefline` suppresses plotting the reference line.

Scatter

`plot#opts`(*scatter_options*) affects the rendition of classifier #'s false-positive rate and ROC scatter points. This option applies only to non-ROC covariate estimation graphing. See [G-2] **graph twoway scatter**.

Reference line

`rlopts`(*cline_options*) affects the rendition of the reference line. See [G-3] **cline_options**.

Y axis, X axis, Titles, Legend, Overall

twoway_options are any of the options documented in [G-3] **twoway_options**, excluding `by()`. These include options for titling the graph (see [G-3] **title_options**) and options for saving the graph to disk (see [G-3] **saving_option**).

boot_options

Bootstrap

`bfile`(*filename*) uses bootstrap replicates of parameters from `rocreg` stored in *filename* to estimate standard errors and confidence intervals of predictions. `bfile()` must be specified with either `roc()` or `invroc()` if parametric estimation with bootstrapping was used.

`btype`(*n* | *p* | *bc*) indicates the desired type of confidence interval rendering. *n* draws normal-based, *p* draws percentile, and *bc* draws bias-corrected confidence intervals for specified false-positive rates and ROC values in `roc()` and `invroc()`. The default is `btype(n)`.

Remarks and examples

[stata.com](http://www.stata.com)

Remarks are presented under the following headings:

Plotting covariate-specific ROC curves
Plotting marginal ROC curves

Plotting covariate-specific ROC curves


The `rocregplot` command is also demonstrated in [R] **rocreg**. We will further demonstrate its use with several examples. Particularly, we will show how `rocregplot` can draw the ROC curves of covariate models that have been fit using `rocreg`.

▷ Example 1: Parametric ROC

In [example 6](#) of [R] [rocreg](#), we fit a probit ROC model to audiology test data from [Norton et al. \(2000\)](#). The estimating equation method of [Alonzo and Pepe \(2002\)](#) was used to fit the model. Gender and age were covariates that affected the control distribution of the classifier `y1` (DPOAE 65 at 2 kHz). Age was a ROC covariate for the model, so we fit separate ROC curves at each age.

Following [Janes, Longton, and Pepe \(2009\)](#), we draw the ROC curves for ages 30, 40, and 50 months. The `at1()`, `at2()`, and `at3()` options are used to specify the age covariates.

```
. use https://www.stata-press.com/data/r18/nhhs
(Norton - neonatal audiology data)
. rocreg d y1, probit ctrlcov(currage male) ctrlmodel(linear) roccov(currage)
> cluster(id) bseed(56930) bsave(nhhs2y1, replace)
(output omitted)
. rocregplot, at1(currage=30) at2(currage=40) at3(currage=50)
```



Here we use the default entries of the legend, which indicate the “at #” within the specified `at*` options and the classifier to which the curve corresponds. ROC curve one corresponds with `currage=30`, two with `currage=40`, and three with `currage=50`. The positive effect of age on the ROC curve is evident. At an age of 30 months (`currage=30`), the ROC curve of `y1` (DPOAE 65 at 2 kHz) is nearly equivalent to that of a noninformative test that gives equal probability to hearing loss. At age 50 months (`currage=50`), corresponding to some of the oldest children in the study, the ROC curve shows that test `y1` (DPOAE 65 at 2 kHz) is considerably more powerful than the noninformative test.

You may create your own legend by specifying the `legend()` option. The default legend is designed for the possibility of multiple covariates. Here we could change the legend entries to `currage` values and gain some extra clarity. However, this may not be feasible when there are many covariates present.

◀

We can also use `rocregplot` after maximum likelihood estimation.

▷ Example 2: Maximum likelihood ROC

We return to the audiology study with frequency (`xf`), intensity (`x1`), and hearing loss severity (`xd`) covariates from [Stover et al. \(1996\)](#) that we examined in [example 10](#) of [R] [rocreg](#). Negative

signal-to-noise ratio is again used as a classifier. Using maximum likelihood, we fit a probit model to these data with the indicated ROC covariates.

After fitting the model, we wish to compare the ROC curves of two covariate combinations. The first has an intensity value of 5.5 (the lowest intensity, corresponding to 55 decibels) and a frequency of 10.01 (the lowest frequency, corresponding to 1001 hertz). We give the first combination a hearing loss severity value of 0.5 (the lowest). The second covariate combination has the same frequency, but the highest intensity value of 6.5 (65 decibels). We give this second covariate set a higher severity value of 4. We will visually compare the two ROC curves resulting from these two covariate value combinations.

We specify false-positive rates of 0.7 first followed by 0.2 in the `roc()` option to visually compare the size of the ROC curve at large and small false-positive rates. Because maximum likelihood estimation was used to fit the model, a Wald confidence interval is produced for the estimated ROC value and false-positive rate parameters. Further details are found in [Methods and formulas](#).

```
. use https://www.stata-press.com/data/r18/dp
(Stover - DPOAE test data)
. rocreg d nsnr, probit ctrlcov(xf x1) roccov(xf x1 xd) ml cluster(id)
(output omitted)
```

```
. rocregplot, at1(xf=10.01, xl=5.5, xd=.5) at2(xf=10.01, xl=6.5, xd=4) roc(.7)
```

ROC curve


Status : d
Classifier: nsnr

Under covariates:

| | at1 | | | |
|-----|-------------|-----------|----------------------|----------|
| xf | 10.01 | | | |
| xl | 5.5 | | | |
| xd | .5 | | | |
| ROC | Coefficient | Std. err. | [95% conf. interval] | |
| .7 | .978951 | .0097382 | .9598645 | .9980376 |

Under covariates:

| | at2 | | | |
|-----|-------------|-----------|----------------------|----------|
| xf | 10.01 | | | |
| xl | 6.5 | | | |
| xd | 4 | | | |
| ROC | Coefficient | Std. err. | [95% conf. interval] | |
| .7 | .9985001 | .0009657 | .9966073 | 1.000393 |



At the higher false-positive rate value of 0.7, we see little difference in the ROC values and note that the confidence intervals nearly overlap. Now, we view the same curves with the lower false-positive rate compared.

8 rocregplot — Plot marginal and covariate-specific ROC curves after rocreg

```
. rocregplot, at1(xf=10.01, xl=5.5, xd=.5) at2(xf=10.01, xl=6.5, xd=4) roc(.2)
```

ROC curve


```
Status      : d
Classifier: nsnr
```

Under covariates:

| | at1 | | | |
|-----|-------------|-----------|----------------------|----------|
| xf | 10.01 | | | |
| xl | 5.5 | | | |
| xd | .5 | | | |
| ROC | Coefficient | Std. err. | [95% conf. interval] | |
| .2 | .7608593 | .0510501 | .660803 | .8609157 |

Under covariates:

| | at2 | | | |
|-----|-------------|-----------|----------------------|----------|
| xf | 10.01 | | | |
| xl | 6.5 | | | |
| xd | 4 | | | |
| ROC | Coefficient | Std. err. | [95% conf. interval] | |
| .2 | .9499408 | .0179824 | .914696 | .9851856 |



The lower false-positive rate of 0.2 shows clearly distinguishable ROC values. Now, we specify option `invroc(.5)` to view how the false-positive rates vary at a ROC value of 0.5.


```
. rocregplot, at1(xf=10.01, xl=5.5, xd=.5) at2(xf=10.01, xl=6.5, xd=4) invroc(.5)
```

False-positive rate


```
Status      : d
Classifier: nsnr
```

Under covariates:

| | at1 | | | |
|--------|-------------|-----------|----------------------|----------|
| xf | 10.01 | | | |
| xl | 5.5 | | | |
| xd | .5 | | | |
| invROC | Coefficient | Std. err. | [95% conf. interval] | |
| .5 | .0578036 | .0198626 | .0188736 | .0967336 |

Under covariates:

| | at2 | | | |
|--------|-------------|-----------|----------------------|----------|
| xf | 10.01 | | | |
| xl | 6.5 | | | |
| xd | 4 | | | |
| invROC | Coefficient | Std. err. | [95% conf. interval] | |
| .5 | .0055624 | .0032645 | -.0008359 | .0119607 |



At a ROC value of 0.5, the false-positive rates for both curves are small and close to one another. ◀

□ Technical note

We can use the `testnl` command to support our visual observations with statistical inference. We use it to perform a Wald test of the null hypothesis that the two ROC curves just rendered are equal at a false-positive rate of 0.7.

```

. testnl normal(_b[i_cons]+10.01*_b[xf]+5.5*_b[x1]
> +.5*_b[xd]+_b[s_cons]*invnormal(.7)) =
> normal(_b[i_cons]+10.01*_b[xf]+6.5*_b[x1]
> + 4*_b[xd]+_b[s_cons]*invnormal(.7))
(1) normal(_b[i_cons]+10.01*_b[xf]+5.5*_b[x1] +.5*_b[xd]+_b[s_cons]*invnormal(.7)) =
normal(_b[i_cons]+10.01*_b[xf]+6.5*_b[x1] + 4*_b[xd]+_b[s_cons]*invnormal(.7))
      chi2(1) =      4.53
      Prob > chi2 =    0.0332

```

The test is significant at the 0.05 level, and thus we find that the two curves are significantly different. Now, we will use `testnl` again to test equality of the false-positive rates for each curve with a ROC value of 0.5. The inverse ROC formula used is derived in *Methods and formulas*.

```

. testnl normal((invnormal(.5)-(_b[i_cons]+10.01*_b[xf]+5.5*_b[x1]+.5*_b[xd]))
> /_b[s_cons]) =
> normal((invnormal(.5)-(_b[i_cons]+10.01*_b[xf]+6.5*_b[x1]+4*_b[xd]))
> /_b[s_cons])
(1) normal((invnormal(.5)-(_b[i_cons]+10.01*_b[xf]+5.5*_b[x1]+.5*_b[xd]))
> /_b[s_cons]) = normal((invnormal(.5)-(_b[i_cons]+10.01*_b[xf]+6.5*_b[x1]
> +4*_b[xd])) /_b[s_cons])
      chi2(1) =      8.01
      Prob > chi2 =    0.0046

```

We again reject the null hypothesis that the two curves are equal at the 0.05 level. □

The model of our last example was also fit using the estimating equations method in [example 7](#) of [\[R\] rocreg](#). We will demonstrate `rocgplot` after that model fit as well.

► Example 3: Parametric ROC, invROC, and ROC value

In [example 2](#), we used `rocgplot` after a maximum likelihood model fit of the ROC curve for classifier `nsnr` and covariates frequency (`xf`), intensity (`x1`), and hearing loss severity (`xd`). The data were obtained from the audiology study described in [Stover et al. \(1996\)](#). In [example 7](#) of [\[R\] rocreg](#), we fit the model using the estimating equations method of [Alonzo and Pepe \(2002\)](#). Under this method, bootstrap resampling is used to make inferences. We saved 50 bootstrap replications in `nsnrf.dta`, which we re-create below.

We use `rocgplot` to draw the ROC curves for `nsnr` under the covariate values `xf = 10.01`, `x1 = 5.5`, and `xd = .5`, and `xf = 10.01`, `x1 = 6.5`, and `xd = 4`. The `at#()` options are used to specify the covariate values. The previous bootstrap results are made available to `rocgplot` with the `bfile()` option. As before, we will specify 0.2 and 0.7 as false-positive rates in the `roc()` option and 0.5 as a ROC value in the `invroc()` option. We do not specify `btype()` and thus our graph will contain normal-based bootstrap confidence bands, the default.

```
. use https://www.stata-press.com/data/r18/dp
(Stover - DPOAE test data)
. rocreg d nsnr, probit ctrlcov(xf xl) roccov(xf xl xd) cluster(id)
> nobstrata ctrlfprall bseed(156385) breps(50) bsave(nsnrf, replace)
(output omitted)
. rocregplot, at1(xf=10.01, xl=5.5, xd=.5) at2(xf=10.01, xl=6.5, xd=4)
> roc(.7) bfile(nsnrf)
```

ROC curve

Status : d
Classifier: nsnr

Under covariates:

| | at1 |
|----|-------|
| xf | 10.01 |
| xl | 5.5 |
| xd | .5 |

(Replications based on 208 clusters in id)


| ROC | Observed coefficient | Bias | Bootstrap std. err. | [95% conf. interval] | |
|-----|----------------------|----------|---------------------|----------------------|---------------|
| .7 | .9835816 | .0026679 | .0133583 | .9573998 | 1.009763 (N) |
| | | | | .9475966 | .9983087 (P) |
| | | | | .9453403 | .9983087 (BC) |

Under covariates:

| | at2 |
|----|-------|
| xf | 10.01 |
| xl | 6.5 |
| xd | 4 |

(Replications based on 208 clusters in id)

| ROC | Observed coefficient | Bias | Bootstrap std. err. | [95% conf. interval] | |
|-----|----------------------|----------|---------------------|----------------------|---------------|
| .7 | .999428 | .0001748 | .0007784 | .9979024 | 1.000954 (N) |
| | | | | .9975154 | .9999899 (P) |
| | | | | .9965358 | .9999899 (BC) |



As shown in the graph, we find that the ROC values at a false-positive rate of 0.7 are close together, as they were in the maximum likelihood estimation in [example 2](#). We now repeat this process for the lower false-positive rate of 0.2 by using the `roc(.2)` option.

```
. rocregplot, at1(xf=10.01, xl=5.5, xd=.5) at2(xf=10.01, xl=6.5, xd=4)
> roc(.2) bfile(nsnrf)
```

```
ROC curve
      Status   : d
      Classifier: nsnrf
```

Under covariates:

| | at1 |
|----|-------|
| xf | 10.01 |
| xl | 5.5 |
| xd | .5 |

(Replications based on 208 clusters in id)


| ROC | Observed coefficient | Bias | Bootstrap std. err. | [95% conf. interval] | |
|-----|----------------------|----------|---------------------|----------------------|---------------|
| .2 | .7652956 | -.006687 | .0624187 | .6429572 | .8876341 (N) |
| | | | | .6654987 | .8837934 (P) |
| | | | | .6444601 | .8837934 (BC) |

Under covariates:

| | at2 |
|----|-------|
| xf | 10.01 |
| xl | 6.5 |
| xd | 4 |

(Replications based on 208 clusters in id)

| ROC | Observed coefficient | Bias | Bootstrap std. err. | [95% conf. interval] | |
|-----|----------------------|-----------|---------------------|----------------------|---------------|
| .2 | .9672505 | -.0015124 | .0162785 | .9353452 | .9991558 (N) |
| | | | | .9372331 | .9976121 (P) |
| | | | | .9218445 | .9926328 (BC) |



The ROC values are slightly higher at the false-positive rate of 0.2 than they were in the maximum likelihood estimation in [example 2](#). To see if the false-positive rates differ at a ROC value of 0.5, we specify the `invroc(.5)` option.

```
. rocregplot, at1(xf=10.01, xl=5.5, xd=.5) at2(xf=10.01, xl=6.5, xd=4)
> invroc(.5) bfile(nsnrf)
```

False-positive rate

Status : d
Classifier: nsnr

Under covariates:

| | at1 |
|----|-------|
| xf | 10.01 |
| xl | 5.5 |
| xd | .5 |

(Replications based on 208 clusters in id)


| invROC | Observed coefficient | Bias | Bootstrap std. err. | [95% conf. interval] | |
|--------|----------------------|----------|---------------------|----------------------|--------------|
| .5 | .0615144 | .0012744 | .0209516 | .0204499 | .1025789 (N) |
| | | | | .0234184 | .09672 (P) |
| | | | | .0234184 | .09672 (BC) |

Under covariates:

| | at2 |
|----|-------|
| xf | 10.01 |
| xl | 6.5 |
| xd | 4 |

(Replications based on 208 clusters in id)

| invROC | Observed coefficient | Bias | Bootstrap std. err. | [95% conf. interval] | |
|--------|----------------------|----------|---------------------|----------------------|---------------|
| .5 | .0043298 | -.000139 | .003835 | -.0031867 | .0118463 (N) |
| | | | | .0002028 | .0136293 (P) |
| | | | | .0012615 | .0192129 (BC) |




The point estimates of the ROC value and false-positive rate are both computed directly using the point estimates of the ROC coefficients. Calculation of the standard errors and confidence intervals is slightly more complicated. Essentially, we have stored a sample of our ROC covariate coefficient estimates in `nsnrf.dta`. We then calculate the ROC value or false-positive rate estimates using each set of coefficient estimates, resulting in a sample of point estimates. Then, the bootstrap standard error and confidence intervals are calculated based on these bootstrap samples. Details of the computation of the standard error and percentile confidence intervals can be found in [Methods and formulas](#) and in [R] [bootstrap](#).

As mentioned in [R] [rocreg](#), 50 resamples is a reasonable lower bound for obtaining bootstrap standard errors (Mooney and Duval 1993). However, it may be too low for obtaining percentile and bias-corrected confidence intervals. Normal-based confidence intervals are valid when the bootstrap distribution exhibits normality. See [R] [bootstrap postestimation](#) for more details.

We can assess the normality of the bootstrap distribution by using a normal probability plot. Stata provides this in the `pnorm` command (see [R] [Diagnostic plots](#)). We will use `nsnrf.dta` to draw a normal probability plot for the ROC estimate corresponding to a false-positive rate of 0.2. We use the covariate values `xf = 10.01`, `x1 = 6.5`, and `xd = 4`.

```
. use nsnrf
(bootstrap: rocregstat)
. generate double rocp2 = nsnr_b_i_cons + 10.01*nsnr_b_xf + 6.5*nsnr_b_x1 +
> 4*nsnr_b_xd+nsnr_b_s_cons*invnormal(.2)
. replace rocp2 = normal(rocp2)
(50 real changes made)
```

```
. pnorm rocp2
```



The closeness of the points to the horizontal line on the normal probability plot shows us that the bootstrap distribution is approximately normal. So it is reasonable to use the normal-based confidence intervals for ROC at a false-positive rate of 0.2 under covariate values $xf = 10.01$, $x1 = 6.5$, and $xd = 4$.

◀

Plotting marginal ROC curves

The `rocregplot` command can also be used after fitting models with no covariates. We will demonstrate this with an empirical ROC model fit in [R] `rocreg`.


► Example 4: Nonparametric ROC

We run `rocregplot` after fitting the single-classifier, empirical ROC model shown in [example 1](#) of [R] `rocreg`. There we empirically predicted the ROC curve of the classifier rating for the true status variable disease from the [Hanley and McNeil \(1982\)](#) data. The `rocreg` command saves variables `_roc_rating` and `_fpr_rating`, which give the ROC values and false-positive rates, respectively, for every value of `rating`. These variables are used by `rocregplot` to render the ROC curve.

```
. use https://www.stata-press.com/data/r18/hanley, clear
(Tomographic images)
. rocreg disease rating, noboot
Nonparametric ROC estimation                               Number of obs = 109
Control standardization: empirical
ROC method          : empirical
Area under the ROC curve
  Status      : disease
  Classifier: rating
```

| AUC | Observed coefficient | Bias | Bootstrap std. err. | [95% conf. interval] |
|-----|----------------------|------|---------------------|----------------------|
| | .8407708 | . | . | . (N) |
| | | | | . (P) |
| | | | | . (BC) |

```
. rocregplot
```



4

We end our discussion of `rocregplot` by showing its use after a marginal probit model.

► Example 5: Maximum likelihood ROC, invROC, and ROC value

In [example 13](#) of [\[R\] rocreg](#), we fit a maximum-likelihood probit model to each classifier of the fictitious dataset generated from [Hanley and McNeil \(1983\)](#).

We use `rocregplot` after the original `rocreg` command to draw the ROC curves for classifiers `mod1` and `mod3`. This is accomplished by specifying the two variables in the `classvars()` option. We will use the `roc()` option to obtain confidence intervals for ROC values at false-positive rates of 0.15 and 0.75. We will specify the `invroc()` option to obtain false-positive rate confidence intervals for a ROC value of 0.8. As mentioned previously, these are Wald confidence intervals.

First, we will view results for a false-positive rate of 0.75.

```
. use https://www.stata-press.com/data/r18/ct2, clear
(Reconstruction of CT images)
. rocreg status mod1 mod2 mod3, probit ml
(output omitted)
. rocregplot, classvars(mod1 mod3) roc(.75)
```


ROC curve

```
Status : status
Classifier: mod1
```

| ROC | Coefficient | Std. err. | [95% conf. interval] | |
|-----|-------------|-----------|----------------------|----------|
| .75 | .9931655 | .0069689 | .9795067 | 1.006824 |

```
Status : status
Classifier: mod3
```

| ROC | Coefficient | Std. err. | [95% conf. interval] | |
|-----|-------------|-----------|----------------------|----------|
| .75 | .9953942 | .0043435 | .9868811 | 1.003907 |



We see that the estimates for each of the two ROC curves are close. Because this is a marginal model, the actual false-positive rate and the true-positive rate for each observation are plotted in the graph. The added point estimates of the ROC value at false-positive rate 0.75 are shown as diamond (mod3) and circle (mod1) symbols in the upper-right-hand corner of the graph at FPR = 0.75. Confidence bands are also plotted at FPR = 0.75 but are so narrow that they are barely noticeable. Under both classifiers, the ROC value at 0.75 is very high. Now, we will compare these results to those with a lower false-positive rate of 0.15.

```
. rocregplot, classvars(mod1 mod3) roc(.15)
```


ROC curve

```
Status : status
Classifier: mod1
```

| ROC | Coefficient | Std. err. | [95% conf. interval] | |
|-----|-------------|-----------|----------------------|----------|
| .15 | .7934935 | .0801363 | .6364292 | .9505578 |

```
Status : status
Classifier: mod3
```

| ROC | Coefficient | Std. err. | [95% conf. interval] | |
|-----|-------------|-----------|----------------------|----------|
| .15 | .8888596 | .0520118 | .7869184 | .9908008 |



The ROC value for the false-positive rate of 0.15 is more separated in the two classifiers. Here we see that mod3 has a larger ROC value than mod1 for this false-positive rate, but the confidence intervals of the estimates overlap.

By specifying `invroc(.8)`, we obtain `invROC` confidence intervals corresponding to a ROC value of 0.8.

```
. rocregplot, classvars(mod1 mod3) invroc(.8)
```


```
False-positive rate
```

```
Status      : status
Classifier: mod1
```

| invROC | Coefficient | Std. err. | [95% conf. interval] | |
|--------|-------------|-----------|----------------------|----------|
| .8 | .1556435 | .069699 | .019036 | .2922509 |

```
Status      : status
Classifier: mod3
```

| invROC | Coefficient | Std. err. | [95% conf. interval] | |
|--------|-------------|-----------|----------------------|----------|
| .8 | .0661719 | .045316 | -.0226458 | .1549896 |



For estimation of the false-positive rate at a ROC value of 0.8, the confidence intervals overlap. Both classifiers require only a small false-positive rate to achieve a ROC value of 0.8.

◀

Methods and formulas

Details on computation of the nonparametric ROC curve and the estimation of the parametric ROC curve model coefficients can be found in [R] [rocreg](#). Here we describe how to estimate the ROC values and false-positive rates of a parametric model. The cumulative distribution function g can be the standard normal cumulative distribution function.

Methods and formulas are presented under the following headings:

Parametric model: Summary parameter definition
Maximum likelihood estimation
Estimating equations estimation

Parametric model: Summary parameter definition

Conditioning on covariates \mathbf{x} , we have the following ROC curve model:

$$\text{ROC}(u) = g\{\mathbf{x}'\boldsymbol{\beta} + \alpha g^{-1}(u)\}$$

\mathbf{x} can be constant, and $\boldsymbol{\beta} = \beta_0$, the constant intercept.

With simple algebra, we can solve this equation to obtain the false-positive rate value u for a ROC value of r :

$$u = g\left[\{g^{-1}(r) - \mathbf{x}'\boldsymbol{\beta}\}\alpha^{-1}\right]$$

Maximum likelihood estimation

We allow maximum likelihood estimation under probit parametric models, so $g = \Phi$. The ROC value and false-positive rate parameters all have closed-form expressions in terms of the covariate values \mathbf{x} , coefficient vector β , and slope parameter α . Thus to estimate these two types of summary parameters, we use the delta method (Oehlert 1992; Phillips and Park 1988). Particularly, we use the `nlcom` command (see [R] `nlcom`) to implement the delta method.

Under maximum likelihood estimation, the coefficient estimates $\hat{\beta}$ and slope estimate $\hat{\alpha}$ are asymptotically normal with variance matrix \mathbf{V} . For convenience, we rename the parameter vector $[\beta', \alpha']$ to the k -parameter vector $\theta = [\theta_1, \dots, \theta_k]$. We will also explicitly refer to the conditioning of the ROC curve by θ in its mention as $\text{ROC}(t, \theta)$.

Under the delta method, the continuous scalar function of the estimate $\hat{\theta}$, $f(\hat{\theta})$ has asymptotic mean $f(\theta)$ and asymptotic covariance

$$\widehat{\text{Var}} \left\{ f(\hat{\theta}) \right\} = \mathbf{fVf}'$$

where \mathbf{f} is the $1 \times k$ matrix of derivatives for which

$$\mathbf{f}_{1j} = \frac{\partial f(\theta)}{\partial \theta_j} \quad j = 1, \dots, k$$

The asymptotic covariance of $f(\hat{\theta})$ is estimated and then used in conjunction with $f(\hat{\theta})$ for further inference, including Wald confidence intervals, standard errors, and hypothesis testing.

Estimating equations estimation

When we fit a model using the Alonzo and Pepe (2002) estimating equations method, we use the bootstrap to perform inference on the ROC curve summary parameters. Each bootstrap sample provides a sample of the coefficient estimates β and the slope estimates α . Using the formulas above, we can obtain an estimate of the ROC value or false-positive rate for each resample.

By making these calculations, we obtain a bootstrap sample of our summary parameter estimate. We then obtain bootstrap standard errors, normal approximation confidence intervals, percentile confidence intervals, and bias-corrected confidence intervals using this bootstrap sample. Further details can be found in [R] `bootstrap`.

References

- Alonzo, T. A., and M. S. Pepe. 2002. Distribution-free ROC analysis using binary regression techniques. *Biostatistics* 3: 421–432. <https://doi.org/10.1093/biostatistics/3.3.421>.
- Bamber, D. 1975. The area above the ordinal dominance graph and the area below the receiver operating characteristic graph. *Journal of Mathematical Psychology* 12: 387–415. [https://doi.org/10.1016/0022-2496\(75\)90001-2](https://doi.org/10.1016/0022-2496(75)90001-2).
- Choi, B. C. K. 1998. Slopes of a receiver operating characteristic curve and likelihood ratios for a diagnostic test. *American Journal of Epidemiology* 148: 1127–1132. <https://doi.org/10.1093/oxfordjournals.aje.a009592>.
- Hanley, J. A., and B. J. McNeil. 1982. The meaning and use of the area under a receiver operating characteristic (ROC) curve. *Radiology* 143: 29–36. <https://doi.org/10.1148/radiology.143.1.7063747>.
- . 1983. A method of comparing the areas under receiver operating characteristic curves derived from the same cases. *Radiology* 148: 839–843. <https://doi.org/10.1148/radiology.148.3.6878708>.

- Janes, H., G. M. Longton, and M. S. Pepe. 2009. Accommodating covariates in receiver operating characteristic analysis. *Stata Journal* 9: 17–39.
- Lora, D., I. Contador, J. F. Pérez-Regadera, and A. Gómez de la Cámara. 2016. Features of the area under the receiver operating characteristic (ROC) curve. A good practice. *Stata Journal* 16: 185–196.
- Mooney, C. Z., and R. D. Duval. 1993. *Bootstrapping: A Nonparametric Approach to Statistical Inference*. Newbury Park, CA: Sage.
- Norton, S. J., M. P. Gorga, J. E. Widen, R. C. Folsom, Y. Sininger, B. Cone-Wesson, B. R. Vohr, K. Mascher, and K. Fletcher. 2000. Identification of neonatal hearing impairment: Evaluation of transient evoked otoacoustic emission, distortion product otoacoustic emission, and auditory brain stem response test performance. *Ear and Hearing* 21: 508–528. <https://doi.org/10.1097/00003446-200010000-00013>.
- Oehlert, G. W. 1992. A note on the delta method. *American Statistician* 46: 27–29. <https://doi.org/10.2307/2684406>.
- Phillips, P. C. B., and J. Y. Park. 1988. On the formulation of Wald tests of nonlinear restrictions. *Econometrica* 56: 1065–1083. <https://doi.org/10.2307/1911359>.
- Stover, L., M. P. Gorga, S. T. Neely, and D. Montoya. 1996. Toward optimizing the clinical utility of distortion product otoacoustic emission measurements. *Journal of the Acoustical Society of America* 100: 956–967. <https://doi.org/10.1121/1.416207>.

Also see

[R] **rocreg** — Receiver operating characteristic (ROC) regression

[R] **rocreg postestimation** — Postestimation tools for rocreg

[U] **20 Estimation and postestimation commands**

Stata, Stata Press, and Mata are registered trademarks of StataCorp LLC. Stata and Stata Press are registered trademarks with the World Intellectual Property Organization of the United Nations. StataNow and NetCourseNow are trademarks of StataCorp LLC. Other brand and product names are registered trademarks or trademarks of their respective companies. Copyright © 1985–2023 StataCorp LLC, College Station, TX, USA. All rights reserved.



For suggested citations, see the FAQ on [citing Stata documentation](#).